Sov. J. Numer. Anal. Math. Modelling, Vol.6, No.4, pp. 315–324 (1991)

On formulae for roots of cubic equation

V. I. LEBEDEV

Abstract – A universal formula is derived for roots of a third-degree general algebraic equation

$$ax^3 + bx^2 + cx + d = 0$$

with either complex or real coefficients. This formula has the same trigonometric form both for complex and real roots. The explicit expressions for parameters of this formula for the case of complex coefficients are obtained. Derivation of formulae is based on representating the general cubic polynomial in terms of the cubic Chebyshev polynomial of the first kind and explicit expressions for its roots. The parameters of this representation are also determined. Furthermore, we give the FORTRAN programs for computing the roots of third- and fourth-degree equations with real coefficients, in which the formulae obtained are realized.

Starting with investigations carried out by Italian mathematicians Del Ferro and Tartali [8] the historical course of developments has been such that the methods for deriving formulae for roots of the cubic equation

$$P_3(x) = 0 \tag{0.1}$$

where

$$P_3(x) = ax^3 + bx^2 + cx + d \tag{0.2}$$

and $a \neq 0$, b, c, d are complex numbers, are based on the successive utilization of two types of transformations. First, by changing the variable $x = y - b/(3a)$ equation (0.1) is transformed into the incomplete cubic equation, and then the resulting equation is transformed using auxiliary variables. This approach has led to the classical Tartali–Cardano and Vietto formulae in the trigonometric form in the so-called irreducible case. For determining the required branches of radicals for real coefficients and complex roots the Tartali–Cardano formulae involve trigonometric and hyperbolic functions of an argument z, where the quantity $3z$ is sought as a solution to an equation involving the similar function [1,4,7,10]. Such an approach leads to further complexity of computations.

It turns out that there exists a transformation of the complete cubic equation which consists in expressing polynomial (0.2) in terms of the Chebyshev polynomial. After this transformation the roots of equation (0.1) can be computed directly and uniquely by a universal and simple algorithm. Here we indicate the papers where similar transformations were applied to the incomplete cubic equation; the paper by Klein [3] dealt with the reduction to the dihedron equation, and the rational transformation was used in [6]. In this paper we derive a universal formula for computing the roots of the third-degree general equation with complex or real coefficients. It has a uniquely determined trigonometric form for both real and complex roots. We also derive explicit expressions for parameters of this formula in the case of complex coefficients.

1. FORMULAE FOR ROOTS OF EQUATION WITH CHEBYSHEV POLYNOMIAL

We first recall the formulae for computing the roots of the equation of the form

$$T_n(y) = A \tag{1.1}$$

where $T_n(y)$ is the nth degree Chebyshev polynomial of the first kind, and A is an arbitrary complex number. To this end, in equation (1.1) we make the change of variable

$$y = \tfrac{1}{2}(z + z^{-1}). \tag{1.2}$$

Then [9] $T_n(y) = \frac{1}{2}(z^n + z^{-n})$, and equation (1.1) is replaced with the equation

$$z^{2n} - 2Az^n + 1 = 0. \tag{1.3}$$

Zhukovsky transformation (1.2) maps the domain $D = \{z: |z| \geqslant 1\}$ of the complex plane Z onto the entire complex plane Y with the cut-cross $-1 \leqslant y \leqslant 1$. Let $z \in D$ in transformation (1.2). Then from equation (1.3) we obtain

$$z^n = \psi(A) \tag{1.4}$$

where

$$\psi(z) = z + \sqrt{z^2 - 1} \tag{1.5}$$

and in (1.5) the branch of the function $\psi(z)$ is chosen which is defined by the formula

$$\psi(z) = |z| \exp(\mathrm{i} \arg z) + |z^2 - 1|^{1/2} \exp\{\tfrac{\mathrm{i}}{2}[\arg(z-1) + \arg(z+1)]\}.$$

Under these assumptions the following relation is valid:

$$A + \sqrt{A^2 - 1} = R \exp(\mathrm{i}\varphi) \tag{1.6}$$

where $R \geqslant 1$. Using this relation we derive the formulae for the roots $z_k \in D$ of equation (1.3):

$$z_k = R^{1/n} \exp[\tfrac{\mathrm{i}}{n}(\varphi + 2k\pi)] \tag{1.7}$$

where, for instance, $k = 0, 1, \ldots, n-1$. Making use of change of variable (1.2), we obtain the formulae for the roots y_k of equation (1.1):

$$y_k = \frac{1}{2}\left(R_n^+ \cos\frac{\varphi + 2k\pi}{n} + \mathrm{i} R_n^- \sin\frac{\varphi + 2k\pi}{n}\right), \qquad k = 0, 1, \ldots, n-1 \tag{1.8}$$

where $R_n^{\pm} = R^{1/n} \pm R^{-1/n}$.

Let us derive formulae for computing the quantities R and φ in expression (1.6) for the case where A is a complex number. Let $A = a + b\mathrm{i}$. Then it follows from (1.6) that

$$A = \tfrac{1}{2}[(R + R^{-1})\cos\varphi + \mathrm{i}(R - R^{-1})\sin\varphi]$$

that is

$$\cos\varphi = \frac{2a}{R + R^{-1}}, \qquad \sin\varphi = \frac{2b}{R - R^{-1}}. \tag{1.9}$$

Substituting these expressions into the identity $\cos^2\varphi + \sin^2\varphi \equiv 1$, we obtain the quadratic equation

$$s^2 - 2|A|^2 s - 2(b^2 - a^2) - 1 = 0$$

for the quantity $s = \frac{1}{2}(R^2 + R^{-2})$. The roots of this equation are real and by solving it we find $s > 1$ and $R > 1$:

$$s = |A|^2 + [|A|^4 + 2(b^2 - a^2) + 1]^{1/2}$$
$$R = [s + (s^2 - 1)^{1/2}]^{1/2}. \tag{1.10}$$

Now the quantity φ can be determined by formulae (1.9).

2. TRANSFORMATION OF CUBIC EQUATION AND FORMULAE IN THE TRIGONOMETRIC FORM FOR COMPUTING ITS ROOTS

Let

$$\tau = b^2 - 3ac\,, \qquad q = b(9ac - 2b^2) - 27a^2 d$$
$$\sqrt{\tau} = |\tau|^{1/2}\exp(\tfrac{i}{2}\arg\tau)\,. \tag{2.1}$$

The following statement is valid.

Lemma 2.1. For $a \neq 0$ and $\tau \neq 0$ any third-degree polynomial $P_3(x)$ with complex coefficients can be presented in the form

$$P_3(x) = \alpha T_3(\beta^{-1}(x - \gamma)) - \delta \tag{2.2}$$

where α, β, γ and δ are complex numbers.

Proof. Indeed, taking into account that $T_3(y) = 4y^3 - 3y$ and making equal the coefficients at the same powers of x in representation (2.2), we obtain the following system of equations for α, β, γ and δ:

$$4\alpha\beta^{-3} = a\,, \qquad 12\alpha\beta^{-3}\gamma^2 - 3\alpha\beta^{-1} = c$$

$$-12\alpha\beta^{-3}\gamma = b\,, \qquad -4\alpha\beta^{-3}\gamma^3 + 3\alpha\beta^{-1}\gamma - \delta = d\,.$$

The solution of this system yields

$$\alpha = \frac{2\tau\sqrt{\tau}}{27a^2}, \quad \beta = \frac{2\sqrt{\tau}}{3a}, \quad \gamma = -\frac{b}{3a}, \quad \delta = \frac{q}{27a^2}\,. \tag{2.3}$$

Now setting $n = 3$ and $y = \beta^{-1}(x - \gamma)$ in equation (1.1) and

$$A = \frac{\delta}{\alpha} = \frac{q}{2\tau\sqrt{\tau}} \tag{2.4}$$

denoting

$$\sigma = (3a)^{-1}, \quad R_3^{\pm} = R^{1/3} \pm R^{-1/3} \tag{2.5}$$

and using expression (1.8) we derive the formulae for the roots x_k of equation (0.1) for $a \neq 0$ and $\tau \neq 0$:

$$x_k = \sigma\left[\sqrt{\tau}\left(R_3^+ \cos\frac{\varphi + 2k\pi}{3} + \mathrm{i} R_3^- \sin\frac{\varphi + 2k\pi}{3}\right) - b\right] \tag{2.6}$$

where we can assume that $k = 0, \pm 1$ or $k = 1,2,3$ on the right-hand side. We recall that the quantity $\sqrt{\tau}$ is determined by formulae (1.10) while the quantities $R_3^\pm$ and φ are determined by formulae (2.5) and (2.6). Note that in the case of $\tau = 0$ the roots of equations (0.1) can be readily determined in the following way:

$$x_1 = \sigma(\sqrt[3]{q} - b), \quad x_{2,3} = -\frac{\sigma}{2}[\sqrt[3]{q}(1 \pm \mathrm{i}\sqrt{3}) + b]. \tag{2.7}$$

Here, by $\sqrt[3]{q}$ we mean the value of the branch of the function $\sqrt[3]{z}$ for which $\sqrt[3]{1} = 1$.

Let us investigate first formulae (2.6) in the case of the real coefficients a, b, c and d. By x_1 we will always denote the real root, while the other two roots will be denoted by x_2 and x_3, respectively. Consider the following three cases: $\tau > 0$ in cases 1 and 2 and $\tau < 0$ in case 3.

Case 1. If $-1 \leqslant A \leqslant 1$, then $\tau > 0$, $R = 1$ and $\cos\varphi = A$ (we can assume that $\sin\varphi \geqslant 0$), and all three roots are real and equal to the expressions

$$\sigma\left(2\sqrt{\tau}\cos\frac{\varphi + 2k\pi}{3} - b\right), \quad k = 0, \pm 1 \quad \text{or} \quad k = 1,2,3. \tag{2.8}$$

When denoting

$$c_0 = \cos\frac{\varphi}{3}, \quad p = c_0\sqrt{\tau}, \quad r = \sqrt{3}\sqrt{\tau}\,|1 - c_0^2|^{1/2}$$

we obtain

$$x_1 = \sigma(2p - b), \quad x_2 = \sigma(r - p - b), \quad x_3 = -\sigma(p + r + b). \tag{2.9}$$

When $|A| = 1$, there exist two coinciding roots: $\varphi = 0$ and $x_2 = x_3$ for $A = 1$; $\varphi = \pi$ and $x_1 = x_2$ for $A = -1$.

Case 2. The quantity A is real and $|A| > 1$. Then $\tau > 0$ and $R = |A| + |A^2 - 1|^{1/2}$, $\varphi = 0$ for $A > 1$ and $\varphi = \pi$ for $A < -1$, that is

$$\begin{aligned} x_1 &= \sigma[\operatorname{sign}(A)\sqrt{\tau}R_3^+ - b] \\ x_{2,3} &= \sigma\left[-\operatorname{sign}(A)\frac{\sqrt{\tau}}{2}R_3^+ - b \pm \mathrm{i}\frac{\sqrt{3}}{2}\sqrt{\tau}R_3^-\right]. \end{aligned} \tag{2.10}$$

Case 3. $\tau < 0$.

$$A = B\mathrm{i}, \quad B = -\frac{q}{2\tau\sqrt{|\tau|}}$$

$R = |B| + \sqrt{|B|^2 + 1}$, $\varphi = \pi/2$ for $B > 0$ and $\varphi = -\pi/2$ for $B < 0$ and

$$
\begin{aligned}
x_1 &= \sigma[\operatorname{sign}(B)\sqrt{|\tau|}\,R_3^- - b] \\
x_{2,3} &= \sigma\left[-\operatorname{sign}(B)\frac{\sqrt{|\tau|}}{2}R_3^- - b \pm i\frac{\sqrt{3}}{2}\sqrt{|\tau|}\,R_3^+\right].
\end{aligned}
\tag{2.11}
$$

Therefore, formulae (2.7) and (2.9)–(2.11) cover all cases of the roots of equation (0.1) for $a \neq 0$ and its real coefficients.

In the case of complex coefficients a, b, c and d of equation (0.1), the quantities R and φ in (2.6) can be computed by formulae (1.9) and (1.10).

3. FORTRAN ROUTINES FOR COMPUTING ROOTS OF THIRD- AND FOURTH-DEGREE EQUATIONS

The routine TC computes the roots of cubic equation (0.1) in the case of real coefficients. The calling sequence is

$$\mathrm{TC}(a,b,c,d,x1,x2,x3,L)$$

where a, b, c and d are coefficients of equation (0.1), and L is the number of computed complex roots. In the case of $L=0$ the roots are $x1$, $x2$ and $x3$, and also $x1 \geqslant x2 \geqslant x3$. In the case of $L=2$: $x1$ is a real root, while $x2$ and $x3$ are, respectively, the real and the imaginary parts of the complex conjugate pair of roots.

The routine FERR4 computes by the Ferrari method the roots of the fourth-degree equation of the form

$$x^4 + ax^3 + bx^2 + cx + d = 0 \tag{3.1}$$

with real coefficients using the routine TC. The calling sequence is

$$\mathrm{FERR4}(a,b,c,d,x1,x2,x3,x4,L1)$$

where a, b, c and d are coefficients of equation (3.1) while the quantity $L1$ characterizes the number of complex roots. In the case of $L1=0$ all roots $x1 \geqslant x2 \geqslant x3 \geqslant x4$ are real; in case of $L1=2$: $x1 \geqslant x2$ are real roots, while $x3$ and $x4$ are the real and imaginary parts of the pair of complex roots and in the case of $L1=3$: $x1$ and $x2$ (respectively $x3$ and $x4$) are the real and the imaginary parts of complex roots.

REFERENCE

1. I. N. Bronshtein and K. A. Semendyaev, *A Handbook of Mathematics*. Nauka, Moscow, 1986 (in Russian).
2. D. K. Faddeev, *Lectures in Algebra*. Nauka, Moscow, 1984 (in Russian).
3. F. Klein, *Elementarmethematik Vom Höheren Standpunkte Aus Erster Band*. Arithmetik. Algebra. Analysis. Dritte Auflage. Verlag Von Julius Springer, Berlin, 1924.
4. A. Korn and M. Korn, *Mathematical Handbook for Scientists and Engineers*. McGraw–Hill Book Company, New–York, 1968.
5. L. G. Kurosh, *A Course of Abstract Algebra*. Nauka, Moscow, 1971 (in Russian).
6. E. G. Lyapin and A. E. Evseev, *Algebra and Number Theory*. Prosveschenie, Moscow, 1978 (in Russian).
7. A. P. Mishina and I. V. Proscuryakov, *Abstract Algebra*. Fizmatgis, Moscow, 1962 (in Russian).
8. V. A. Nikiforovsky, *The World of Equations*. Nauka, Moscow, 1987 (in Russian).
9. S. Pashkovsky, *Numerical Application of Chebyshev Polynomials and Series*. Nauka, Moscow, 1983 (in Russian).
10. V. I. Smirnov, *A Course of Higher Mathematics*, vol. 1. Moscow, 1953 (in Russian).

Appendix A. The source code of the TC routine.

```
      SUBROUTINE TC(A,B,C,D,X1,X2,X3,L)
      IMPLICIT REAL*8(A-H,O-Z)
      T=1.732050807568877D0
      S=.3333333333333333D0
      T2=B*B
      T3=3.D0*A
      T4=T3*C
      P=T2-T4
      X3=DABS(P)
      X3=DSQRT(X3)
      X1=B*(T4-P-P)-3.D0*T3*T3*D
      X2=DABS(X1)
      X2=X2**S
      T2=1.D0/T3
      T3=B*T2
      IF(X3.GT.1.D-32*X2) GO TO 1
      IF(X1.LT.0.D0)  X2=-X2
      X1=X2*T2
      X2=-.5D0*X1
      X3=-T*X2
      IF(DABS(X3).GT.1.D-32) GO TO 15
      X3=X2
                              GO TO 2
1     T1=.5D0*X1/(P*X3)
      X2=DABS(T1)
      T2=X3*T2
      T=T*T2
      T4=X2*X2
      IF(P.LT.0.D0)  GO TO 7
      X3=DABS(1.D0-T4)
      X3=DSQRT(X3)
      IF(T4.GT.1.D0) GO TO 5
      T4=DATAN2(X3,T1)*S
      X3=DCOS(T4)
      T4=DSQRT(1.D0-X3*X3)*T
      X3=X3*T2
      X1=X3+X3
      X2=T4-X3
      X3=-(T4+X3)
      IF(X2.GT.X3)    GO TO 2
      T2=X2
      X2=X3
      X3=T2
2     L=0
      IF(X1.GT.X2)    GO TO 3
      T2=X1
      X1=X2
      X2=T2
      IF(T2.GT.X3)    GO TO 3
      X2=X3
      X3=T2
3     X3=X3-T3
                              GO TO 20
5     P=(X2+X3)**S
      T4=1.D0/P
      IF(T1) 11,13,13
```

```
7      P=X2+DSQRT(T4+1.D0)
       P=P**S
       T4=-1.D0/P
       IF(T1.LT.0.D0) GO TO 13
11     T2=-T2
13     X1=(P+T4)*T2
       X2=-.5D0*X1
       X3=.5D0*T*(P-T4)
15     L=2
20     X1=X1-T3
       X2=X2-T3
       RETURN
       END
```

Appendix B. The source code of the FERR4 routine.

```
       SUBROUTINE FERR4(A,B,C,D,X1,X2,X3,X4,L1)
       IMPLICIT REAL*8(A-H,O-Z)
       L1=0
       R9=.5D0*A
       S1=-.5D0*R9
       S2=R9*R9-B
       S3=DABS(D)
       S6=DABS(C)
       S=1.D-32*(S6+DABS(A)+DABS(B)+1.D0)
       IF(S3.GT.S)     GO TO 7
       X1=0.D0
       IF(S6.GT.S)     GO TO 6
       X2=0.D0
       X4=DABS(S2)
       X4=DSQRT(X4)
       IF(S2.LT.0.D0) GO TO 5
       X3=X4-R9
       X4=-X4-R9
                       GO TO 34
 5     X3=-R9
       L1=2
                       GO TO 44
 6     CALL TC(1.D0,A,B,C,X2,X3,X4,L1)
                       GO TO 32
 7     IF(DABS(C+S2*R9).LE.S) GO TO 60
       C1=A*C-4.D0*D
       A1=-4D0*S2*D-C*C
       CALL TC(1.D0,-B,C1,A1,X1,X2,X3,L)
       S0=X1
       J=0
 8     J=J+1
       R7=S2+S0
       S9=.5D0*S0
       S7=DABS(R7)
       S7=DSQRT(S7)
       R8=S9*S9-D
       S8=DABS(R8)
       S8=DSQRT(S8)
       S6=R9*S0-C
       IF(R7.GT.0.D0) GO TO 13
       IF(R8.GT.0.D0) GO TO 13
```

```
      IF(L.NE.0)      GO TO 90
      GO TO (9,11,90),J
 9    S0=X2
                      GO TO 8
 11   S0=X3
                      GO TO 8
 13   IF(S6.LT.0.D0) S7=-S7
      C1=-.5D0*(R9+S7)
      R1=S8+S9
      R2=S8-S9
      R3=1.D-32*S3
      IF(S9.GE.0.D0) GO TO 14
      IF(R2.GT.R3) R1=-D/R2
                      GO TO 16
 14   IF(R1.GT.R3) R2=-D/R1
 16   X1=C1*C1-R1
      X2=DABS(X1)
      X2=DSQRT(X2)
      IF(X1.LT.0.D0) GO TO 10
      IF(C1.GE.0.D0) GO TO 17
      X2=C1-X2
      X1=R1/X2
                      GO TO 20
 17   X1=C1+X2
      X2=R1/X1
                      GO TO 20
 10   X1=C1
      L1=1
 20   X3=-.5D0*(R9-S7)
      C1=X3*X3+R2
      X4=DABS(C1)
      B1=DSQRT(X4)
      IF(C1.LT.0.D0) GO TO 30
      IF(X3.LE.0.D0) GO TO 18
      X3=X3+B1
      X4=-R2/X3
                      GO TO 32
 18   X4=X3-B1
      X3=-R2/X4
                      GO TO 32
 60   R4=.5D0*S2
      R2=R4*R4-D
      R3=DABS(R2)
      R3=DSQRT(R3)
      R1=S1*S1+R4
      IF(R2.LT.0.D0) GO TO 64
      R2=R1+R3
      R3=R1-R3
      S2=DABS(R2)
      S2=DSQRT(S2)
      S3=DABS(R3)
      S3=DSQRT(S3)
      IF(R3.LT.0.0)   GO TO 62
      X3=S1+S3
      X4=S1-S3
 61   X2=S1-S2
                      GO TO 98
 62   X3=S1
```

```
      X4=S3
      L1=2
      IF(R2.GE.0.D0) GO TO 61
      X1=S1
      X2=S2
      L1=L1+1
                     GO TO 32
64    L1=3
      IF(DABS(R1).GT.S) GO TO 66
      B1=.707106781186547D0*R3
      X1=S1+B1
      X3=S1-B1
                     GO TO 69
66    S2=DSQRT(R1*R1+R2*R2)
      IF(S2.GT.0.D0) GO TO 67
      S2=0.D0
      B1=0.D0
                     GO TO 68
67    R1=DATAN2(R3,R1)*.5D0
      S2=S2*DCOS(R1)
      B1=S2*DSIN(R1)
68    X1=S1+S2
      X3=S1-S2
69    X2=B1
                     GO TO 31
90    IF(S6.GT.0.D0) S7=-S7
      R1=.25D0*(R9*R9-S7*S7)-S9
      R2=S8+S1*S7
      R3=DSQRT(R1*R1+R2*R2)
      IF(R3.GT.S)    GO TO 94
      S2=0.D0
      S3=0.D0
                     GO TO 96
94    R3=DSQRT(R3)
      R1=DATAN2(R2,R1)*.5D0
      S2=R3*DCOS(S1)
      S3=R3*DSIN(S1)
96    R5=.5D0*S7
      X2=R5+S3
      X3=S1-S2
      X4=R5-S3
      L1=3
98    X1=S1+S2
                     GO TO 32
30    L1=L1+2
31    X4=B1
32    IF(L1-1)  34,40,44
34    IF(X1.LT.X3)   GO TO 36
      IF(X2.GT.X3)   GO TO 44
      B1=X2
      X2=X3
      X3=B1
      IF(X3.GT.X4)   GO TO 44
      X3=X4
      X4=B1
                     GO TO 44
36    B1=X3
      X3=X2
```

```
      X2=X1
      X1=B1
      IF(X4.LT.X2)    GO TO 38
      B1=X4
      X4=X3
      X3=X2
      X2=B1
38    IF(X4.LT.X3)    GO TO 44
      B1=X4
      X4=X3
                      GO TO 42
40    B1=X2
      X2=X4
      X4=B1
      L1=2
      B1=X1
      X1=X3
42    X3=B1
44    RETURN
      END
```